A Declarative Semantics for SNOMED CT Expression Constraints

January 26, 2015

Contents

1 Axiomatic Data Types

1.1 Atomic Data Types

This section identifies the atomic data types that are assumed for the rest of this specification, specifically:

- **SCTID** – a SNOMED CT identifier
- **TERM** – a fully specified name, preferred term or synonym for a SNOMED CT Concept
- **REAL** – a real number
- **STRING** – a string literal
- **GROUP** – a role group identifier
- $\mathbb{N}$ – a non-negative integer
- $\mathbb{Z}$ – an integer

We also introduce several synonyms for *SCTID*:

- **SUBJECT** – a *SCTID* that appears in the *sourceId* position of a relationship.
- **ATTRIBUTE** – a *SCTID* that appears in the *typeId* position of a relationship.
- **REFSETID** – a *SCTID* that identifies a reference set

$[SCTID, TERM, REAL, STRING, GROUP]$

$SUBJECT == SCTID$
$ATTRIBUTE == SCTID$
$REFSETID == SCTID$

1.2 Composite Data Types

While we can't fully specify the behavior of the concrete data types portion of the specification at this point, it is still useful to spell out the anticipated behavior on an abstract level.

- **CONCRETEVALUE** – a string, integer or real literal
- **TARGET** – the target of a relationship that is either an SCTID or a CONCRETEVALUE

$$CONCRETEVALUE ::= string\langle\langle STRING\rangle\rangle \mid integer\langle\langle \mathbb{Z}\rangle\rangle \mid real\langle\langle REAL\rangle\rangle$$
$$TARGET ::= object\langle\langle SCTID\rangle\rangle \mid concrete\langle\langle CONCRETEVALUE\rangle\rangle$$

2 The Substrate

A substrate represents the context of an interpretation. A substrate consists of:

- **c** The set of *SCTID*s (concepts) that are considered valid in the context of the substrate. **References to any** *SCTID* **that is not a member of this set MUST be treated as an error.**
- **a** The set of *SCTID*s that are considered to be valid attributes in the context of the substrate. **Reference to any** *ATTRIBUTE* **that is not a member of this set MUST be treated as an error.**
- **r** A set of relationship quads (subject, attribute, target, group)
- **descs** The subsumption (ISA) closure from general to specific (descendants)
- **ancs** The subsumption (ISA) closure from specific to general (ancestors)
- **refset** The reference sets within the context of the substrate whose members are members are concept identifiers (i.e. are in c). While not formally spelled out in this specification, it is assumed that the typical reference set function would be returning a subset of the *refsetId*/*referencedComponentId* tuples represented in one or more RF2 Refset Distribution tables.

Issue: this is currently the simplest possible kind of Substrate; it represents the DNF form of SNOMED CT with concepts only (i.e., *Expressions* and *Expression Libraries* are not supported).

Issue: this is a *flat* representation (fixed to 1 level of nesting corresponding to role grouping) and does not properly handle Concrete Domains (which almost always involve additional nesting

Open questions: While the Core will not contain cycles, and probably NRC Extensions will not, arbitrary extensions (e.g. as a result of handling Expression libraries) may involve cycles (or at least equivalent concepts).

- **Resolution:** A reflexive, symmetric equivalence relationship (*equiv*) was added to the substrate. How it is determined is left unspecified, but a rule

was added saying that an equivalent concept can not be a descendant of its equivalence.

- Is it correct to interpret the transitive closure directly against the substrate relationships (r) set or is there an implication of a reasoner being invoked somewhere, which could potentially render the transitive closure as logically derived from (r). **Answer:** If substrate includes postcoordinated expressions then subsumption would need to be calculated. **Followup:** How does one know whether it includes postcoordinated expressions? How do we express this decision formally?

- Do we want to assert that it is an error to use a non-attribute concept in a attribute position? If not, should we explicitly include a formal definition of the attribute set? **Answer:** Because the ECL doesn't declare this as an error condition, we shouldn't assert in the Z spec. **Resolution:** $a =$ *descs attribute_concept* assertion pulled from the substrate declaration.

2.1 Substrate Components

Quad Relationships in the substrate are represented a 4 element tuples or "Quads", which consist of a subject, attribute, target and role group identifier.

Quad

$s : SUBJECT$
$a : ATTRIBUTE$
$t : TARGET$
$g : GROUP$

We will also need to recognize some "well known" identifiers: the *is_a* attribute, the *zero_group* and *attribute_concept*, the parent of all attributes

$is\_a : ATTRIBUTE$
$zero\_group : GROUP$
$attribute\_concept : SCTID$
$refset\_concept : SCTID$

2.2 Substrate

The formal definition of substrate follows, where c and r are given and the remainder are derived. The expressions below assert that:

1. All subjects and attributes and targets of type *object* in r must be members of the set of concepts, c. All attributes are also members of a.

2. The is_a relation is irreflexive.

3. *childrenOf includes* the function from a concept in c to the set of concepts that are the source of c in an *is_a* relationship in the *zero_group*. Note

that substrates that include post-coordinated expressions may include additional entries in the *childrenOf* function that are not in the relationships table and, instead, are derived through the application of a DL reasoner.

4. *parentsOf includes* the function from a concept c to the set of concepts that are the target of of c in an *is_a* relationship in the *zero_group*. Note that substrates that include post-coordinated expressions may include additional entries in the *parentsOf* function that are not in the relationships table and, instead, are derived through the application of a DL reasoner.

5. *parentsOf* and *childrenOf* are irreflexive.

6. The *equiv* relationship is symmetric and reflexive.

7. The ancestors function, *ancs* ,is the *irreflexive* transitive closure of the *parentsOf* relationship.

8. The descendants function, *descs*, is the *irreflexive* transitive closure of the inverse of the *childrenOf* relationship.

9. The descendants relationship must not contain any concepts that are declared as equivalent to the root.

10. The reference set function is a function from *subset of* c to set of SCTID's in c. A SCTID that is not in the domain of *refset* cannot appear as the target of a *memberOf* function

Substrate

$$
\begin{array}{l}
c : \mathbb{P}\, SCTID \\
r : \mathbb{P}\, Quad \\
a : \mathbb{P}\, SCTID \\
\\
childrenOf : SCTID \pfun \mathbb{P}\, SCTID \\
parentsOf : SCTID \pfun \mathbb{P}\, SCTID \\
\\
descs : SCTID \pfun \mathbb{P}\, SCTID \\
ancs : SCTID \pfun \mathbb{P}\, SCTID \\
equiv : \mathbb{P}(SCTID \times SCTID) \\
refset : REFSETID \pfun \mathbb{P}\, SCTID \\
\hline
\forall\, rel : r \bullet rel.s \in c \land rel.a \in a \land (object^{\sim} rel.t \in SCTID \Rightarrow object^{\sim} rel.t \in c) \\
\forall\, q : r \bullet q.a \in a \land q.a = is\_a \Rightarrow q.s \neq object^{\sim} q.t \\
\forall\, q : r \bullet q.a \in descs\ attribute\_concept \\
\\
\mathrm{dom}\ childrenOf \subseteq c \land \bigcup(\mathrm{ran}\ childrenOf) \subseteq c \\
\forall\, s : c \bullet parentsOf\ s \subseteq \{q : r \mid q.s = s \land q.a = is\_a \land q.g = zero\_group \bullet object^{\sim} q.t\} \\
\forall\, t : c \bullet childrenOf\ t \subseteq \{q : r \mid q.a = is\_a \land q.t = object\ t \land q.g = zero\_group \bullet q.s\} \\
\forall\, c1, c2 : c \bullet c1 \in parentsOf\ c2 \Leftrightarrow c2 \in childrenOf\ c1 \\
\\
\forall\, x : c \bullet (x \mapsto x) \in equiv \\
\forall\, x_1, x_2 : c \bullet x_1 \mapsto x_2 \in equiv \Rightarrow x_2 \mapsto x_1 \in equiv \\
\\
\forall\, s : c \bullet descs\ s = childrenOf\ s \cup \bigcup\{t : childrenOf\ s \bullet descs\ t\} \\
\forall\, t : c \bullet ancs\ t = parentsOf\ t \cup \bigcup\{s : parentsOf\ t \bullet ancs\ s\} \\
\forall\, r : equiv \bullet first\ r \in \mathrm{dom}\ descs \Rightarrow second\ r \notin (descs\ (first\ r)) \\
\\
\mathrm{dom}\ refset \subseteq descs\ refset\_concept \land \bigcup(\mathrm{ran}\ refset) \subseteq c
\end{array}
$$

3 Result

The result of applying a query against a substrate is either a (possibly empty) set of SCTID's or an *ERROR*.

$$
\begin{array}{l}
ERROR ::= unknownOperation \mid unknownConceptReference \mid \\
\qquad\qquad unknownAttribute \mid unknownRefsetId \\
\\
Result ::= ok\langle\langle \mathbb{P}\, SCTID \rangle\rangle \mid error\langle\langle ERROR \rangle\rangle
\end{array}
$$

4 Interpretation of Intermediate Constructs

This section carries the interpretation of the intermediate constructs – the various forms of expressions and their combinations.

4.1 expressionConstraint

```
    expressionConstraint =
( refinedExpressionConstraint / unrefinedExpressionConstraint )
```

4.1.1 Interpretation

The interpretation of `expressionConstraint` the interpretation of the `refinedExpressionConstraint` or the `unrefinedExpressionConstraint`.

$$
\begin{array}{l}
expressionConstraint ::= \\
\quad ecrec\langle\langle refinedExpressionConstraint\rangle\rangle \mid \\
\quad ecurec\langle\langle unrefinedExpressionConstraint\rangle\rangle
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_expressionConstraint : \\
\quad Substrate \to expressionConstraint \to Result \\
\hline
\forall\, ss : Substrate;\ ec : expressionConstraint \bullet i\_expressionConstraint\ ss\ ec = \\
\quad \textbf{if}\ ecrec^{\sim} ec \in refinedExpressionConstraint \\
\quad\quad \textbf{then}\ i\_refinedExpressionConstraint\ ss\ (ecrec^{\sim} ec) \\
\quad \textbf{else if}\ ecurec^{\sim} ec \in unrefinedExpressionConstraint \\
\quad\quad \textbf{then}\ i\_unrefinedExpressionConstraint\ ss\ (ecurec^{\sim} ec) \\
\quad \textbf{else}\ error\ unknownOperation \\
\hline
\end{array}
$$

4.1.2 unrefinedExpressionConstraint

An `unrefinedExpressionConstraint` is either a `compoundExpressionConstraint` or a `simpleExpressionConstraint`

```
unrefinedExpressionConstraint =
    compoundExpressionConstraint / simpleExpressionConstraint
```

$$
\begin{array}{l}
unrefinedExpressionConstraint ::= \\
\quad ucec\langle\langle compoundExpressionConstraint\rangle\rangle \mid \\
\quad usec\langle\langle simpleExpressionConstraint\rangle\rangle
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_unrefinedExpressionConstraint : \\
\quad Substrate \to unrefinedExpressionConstraint \to Result \\
\hline
\forall\, ss : Substrate;\ uec : unrefinedExpressionConstraint \bullet \\
i\_unrefinedExpressionConstraint\ ss\ uec = \\
\quad \textbf{if}\ ucec^{\sim} uec \in compoundExpressionConstraint \\
\quad\quad \textbf{then}\ i\_compoundExpressionConstraint\ ss\ (ucec^{\sim} uec) \\
\quad \textbf{else if}\ usec^{\sim} uec \in simpleExpressionConstraint \\
\quad\quad \textbf{then}\ i\_simpleExpressionConstraint\ ss\ (usec^{\sim} uec) \\
\quad \textbf{else}\ error\ unknownOperation \\
\hline
\end{array}
$$

4.1.3 refinedExpressionConstraint

```
refinedExpressionConstraint =
     unrefinedExpressionConstraint ":" refinement /
     "(" refinedExpressionConstraint ")"
```

The interpretation of `refinedExpressionConstraint` is the intersection of the interpretation of the `unrefinedExpressionConstraint` and the `refinement`, both of which return a set of SCTID's or an error.

The interpretation of the second option adds no value.

$$refinedExpressionConstraint == unrefinedExpressionConstraint \times refinement$$

$$\begin{array}{|l} i\_refinedExpressionConstraint : \\ \quad Substrate \pfun refinedExpressionConstraint \pfun Result \\ \hline \forall\, ss : Substrate;\ rec : refinedExpressionConstraint \bullet \\ i\_refinedExpressionConstraint\ ss\ rec = \\ \quad intersect\ (i\_unrefinedExpressionConstraint\ ss\ (first\ rec))(i\_refinement\ ss\ (second\ rec)) \end{array}$$

4.2 simpleExpressionConstraint

The interpretation of `simpleExpressionConstraint` is the application of an optional constraint operator to the interpretation of `focusConcept`, which returns a set of SCTID's or an error. The interpretation of an error is the error.

```
simpleExpressionConstraint =
     [constraintOperator ] focusConcept
```

$$constraintOperator ::= descendantOrSelfOf \mid descendantOf \mid ancestorOrSelfOf \mid ancestorOf$$
$$simpleExpressionConstraint == constraintOperator[0\,..\,1] \times focusConcept$$

$$\begin{array}{|l} i\_simpleExpressionConstraint : \\ \quad Substrate \pfun simpleExpressionConstraint \pfun Result \\ \hline \forall\, ss : Substrate;\ sec : simpleExpressionConstraint \bullet \\ i\_simpleExpressionConstraint\ ss\ sec = \\ \quad i\_constraintOperator\ ss\ (first\ sec)\ (i\_focusConcept\ ss\ (second\ sec)) \end{array}$$

4.3 compoundExpressionConstraint

The interpretation of a *compoundExpressionConstraint* is the interpretation of its corresponding component.

```
compoundExpressionConstraint = conjunctionExpressionConstraint |
       disjunctionExpressionConstraint | exclusionExpressionConstraint |
        "(" compoundExpressionConstraint ")"
```

$compoundExpressionConstraint ::=$
$\quad cecConj\langle\langle conjunctionExpressionConstraint\rangle\rangle \mid$
$\quad cecDisj\langle\langle disjunctionExpressionConstraint\rangle\rangle \mid$
$\quad cecExc\langle\langle exclusionExpressionConstraint\rangle\rangle$

$i\_compoundExpressionConstraint :$
$\quad Substrate \pfun compoundExpressionConstraint \pfun Result$

$\forall ss : Substrate;\ cec : compoundExpressionConstraint \bullet$
$i\_compoundExpressionConstraint\ ss\ cec =$
$\quad \textbf{if}\ cecConj^{\sim} cec \in conjunctionExpressionConstraint$
$\quad\quad \textbf{then}\ i\_conjunctionExpressionConstraint\ ss\ (cecConj^{\sim} cec)$
$\quad \textbf{else if}\ cecDisj^{\sim} cec \in disjunctionExpressionConstraint$
$\quad\quad \textbf{then}\ i\_disjunctionExpressionConstraint\ ss\ (cecDisj^{\sim} cec)$
$\quad \textbf{else if}\ cecExc^{\sim} cec \in exclusionExpressionConstraint$
$\quad\quad \textbf{then}\ i\_exclusionExpressionConstraint\ ss\ (cecExc^{\sim} cec)$
$\quad \textbf{else}\ error\ unknownOperation$

$i\_compoundExpressionConstraint' :$
$\quad Substrate \pfun compoundExpressionConstraint \pfun Result$

4.4 conjunctionExpressionConstraint

conjunctionExpressionConstraint is interpreted the conjunction (intersection) of the interpretation of two or more *subExpressionConstraints*

```
conjunctionExpressionConstraint =
        subExpressionConstraint 1*(conjunction subExpressionConstraint)
```

$conjunctionExpressionConstraint == subExpressionConstraint \times \mathrm{seq}_1(subExpressionConstraint)$

$i\_conjunctionExpressionConstraint :$
$\quad Substrate \pfun conjunctionExpressionConstraint \pfun Result$

$\forall ss : Substrate;\ cecr : conjunctionExpressionConstraint \bullet$
$i\_conjunctionExpressionConstraint\ ss\ cecr =$
$\quad \textbf{if}\ tail(second\ cecr) = \langle\rangle\ \textbf{then}$
$\quad\quad intersect\ (i\_subExpressionConstraint\ ss\ (first\ cecr))(i\_subExpressionConstraint\ ss\ (head\ (seco$
$\quad \textbf{else}$
$\quad\quad intersect\ (i\_subExpressionConstraint\ ss\ (first\ cecr))(i\_conjunctionExpressionConstraint\ ss\ ((l$

4.5 disjunctionExpressionConstraint

disjunctionExpressionConstraint is interpreted the disjunction (union) of the interpretation of two or more *subExpressionConstraints*

```
disjunctionExpressionConstraint =
      subExpressionConstraint 1*(disjunction subExpressionConstraint)
```

$$disjunctionExpressionConstraint == subExpressionConstraint \times \text{seq}_1(subExpressionConstraint)$$

$$i\_disjunctionExpressionConstraint : Substrate \pfun disjunctionExpressionConstraint \pfun Result$$

$$\forall ss : Substrate;\ cecr : disjunctionExpressionConstraint \bullet$$
$$i\_disjunctionExpressionConstraint\ ss\ cecr =$$
$$\textbf{if } tail(second\ cecr) = \langle\rangle \textbf{ then}$$
$$union\ (i\_subExpressionConstraint\ ss\ (first\ cecr))(i\_subExpressionConstraint\ ss\ (head\ (second$$
$$\textbf{else}$$
$$union\ (i\_subExpressionConstraint\ ss\ (first\ cecr))(i\_disjunctionExpressionConstraint\ ss\ ((head$$

4.6 exclusionExpressionConstraint

exclusionExpressionConstraint is interpreted the result of removing the members of the second *subExpressionConstraint* from the first

```
exclusionExpressionConstraint =
      subExpressionConstraint exclusion subExpressionConstraint
```

$$exclusionExpressionConstraint == subExpressionConstraint \times subExpressionConstraint$$

$$i\_exclusionExpressionConstraint : Substrate \pfun exclusionExpressionConstraint \pfun Result$$

$$\forall ss : Substrate;\ ecr : exclusionExpressionConstraint \bullet$$
$$i\_exclusionExpressionConstraint\ ss\ ecr =$$
$$minus\ (i\_subExpressionConstraint\ ss\ (first\ ecr))\ (i\_subExpressionConstraint\ ss\ (second\ ecr))$$

4.7 subExpressionConstraint

subExpressionConstraint is interpreted as the interpretation of either a *simpleExpressionConstraint* or a *compoundExpressionConstraint*

```
subExpressionConstraint =
      simpleExpressionConstraint | "(" compoundExpressionConstraint ")
```

$$
\begin{array}{l}
subExpressionConstraint ::= \\
\quad secSec\langle\langle simpleExpressionConstraint\rangle\rangle \mid \\
\quad secCec\langle\langle compoundExpressionConstraint\rangle\rangle
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_subExpressionConstraint : \\
\qquad Substrate \pfun subExpressionConstraint \pfun Result \\
\hline
\forall\, ss : Substrate;\ sec : subExpressionConstraint \bullet \\
i\_subExpressionConstraint\ ss\ sec = \\
\quad \textbf{if}\ (secSec^{\sim} sec) \in simpleExpressionConstraint \\
\qquad \textbf{then}\ i\_simpleExpressionConstraint\ ss\ (secSec^{\sim} sec) \\
\quad \textbf{else if}\ secCec^{\sim} sec \in compoundExpressionConstraint \\
\qquad \textbf{then}\ i\_compoundExpressionConstraint'\ ss\ (secCec^{\sim} sec) \\
\quad \textbf{else}\ error\ unknownOperation \\
\hline
\end{array}
$$

4.8 refinement

`refinement` is the interpretation of one or more `attribute` or `attributeGroup`s joined by `binaryOperator`s

```
binaryOperator = conjunction / disjunction / minus
refinement = (attribute / attributeGroup / "(" refinement ")")
         [binaryOperator  refinement]
```

$$
\begin{array}{l}
refTarget ::= att\langle\langle attribute\rangle\rangle \mid attg\langle\langle attributeGroup\rangle\rangle \\
refinement == refTarget \times \mathrm{seq}(binaryOperator \times refTarget)
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_refTarget : Substrate \to refTarget \to Result \\
i\_refinement : Substrate \to refinement \to Result \\
\hline
\forall\, ss : Substrate;\ rt : refTarget \bullet i\_refTarget\ ss\ rt = \\
\quad \textbf{if}\ att^{\sim} rt \in attribute \\
\qquad \textbf{then}\ i\_attribute\ ss\ (att^{\sim} rt) \\
\quad \textbf{else if}\ attg^{\sim} rt \in attributeGroup \\
\qquad \textbf{then}\ i\_attributeGroup\ ss\ (attg^{\sim} rt) \\
\quad \textbf{else}\ error\ unknownOperation \\
\\
\forall\, ss : Substrate;\ ref : refinement \bullet \\
\quad i\_refinement\ ss\ ref = \\
\quad evalRefinement\ ss\ (i\_refTarget\ ss\ (first\ ref))\ (second\ ref) \\
\hline
\end{array}
$$

$$
\begin{array}{|l}
\hline
evalRefinement : \\
\qquad Substrate \to Result \to \mathrm{seq}(binaryOperator \times refTarget) \to Result \\
\hline
\forall\, ss : Substrate;\ r : Result;\ opt : \mathrm{seq}(binaryOperator \times refTarget) \bullet \\
\quad evalRefinement\ ss\ r\ opt = \\
\qquad \textbf{if}\ opt = \langle\rangle \\
\qquad\quad \textbf{then}\ r \\
\qquad \textbf{else if}\ first\,(head\ opt) = conjunction \\
\qquad\quad \textbf{then}\ evalRefinement\ ss\,(intersect\ r(i\_refTarget'\ ss\,(second\,(head\ opt))))\,(tail\ opt) \\
\qquad \textbf{else if}\ first\,(head\ opt) = disjunction \\
\qquad\quad \textbf{then}\ evalRefinement\ ss\,(union\ r(i\_refTarget'\ ss\,(second\,(head\ opt))))\,(tail\ opt) \\
\qquad \textbf{else if}\ first\,(head\ opt) = exclusion \\
\qquad\quad \textbf{then}\ evalRefinement\ ss\,(minus\ r(i\_refTarget'\ ss\,(second\,(head\ opt))))\,(tail\ opt) \\
\qquad \textbf{else}\ error\ unknownOperation \\
\hline
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_refTarget' : Substrate \to refTarget \to Result \\
\hline
\end{array}
$$

4.9 expressionConstraintValue

```
expressionConstraintValue = simpleExpressionConstraint /
       "(" (refinedExpressionConstraint /
       compoundExpressionConstraint) ")"
```

$$
\begin{array}{l}
expressionConstraintValue ::= ecvsec\langle\langle simpleExpressionConstraint\rangle\rangle \mid \\
\qquad ecvrec\langle\langle refinedExpressionConstraint'\rangle\rangle \mid \\
\qquad ecvcec\langle\langle compoundExpressionConstraint\rangle\rangle \\
[refinedExpressionConstraint']
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_expressionConstraintValue : Substrate \to \\
\qquad expressionConstraintValue \to Result \\
\hline
\forall\, ss : Substrate;\ ecv : expressionConstraintValue \bullet \\
i\_expressionConstraintValue\ ss\ ecv = \\
\quad \textbf{if}\ ecvsec^{\sim} ecv \in simpleExpressionConstraint \\
\qquad \textbf{then}\ i\_simpleExpressionConstraint\ ss\ (ecvsec^{\sim} ecv) \\
\quad \textbf{else if}\ ecvrec^{\sim} ecv \in refinedExpressionConstraint' \\
\qquad \textbf{then}\ i\_refinedExpressionConstraint'\ ss\ (ecvrec^{\sim} ecv) \\
\quad \textbf{else if}\ ecvcec^{\sim} ecv \in compoundExpressionConstraint \\
\qquad \textbf{then}\ i\_compoundExpressionConstraint\ ss\ (ecvcec^{\sim} ecv) \\
\quad \textbf{else}\ error\ unknownOperation \\
\hline
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_refinedExpressionConstraint' : \\
\qquad Substrate \nrightarrow refinedExpressionConstraint' \nrightarrow Result \\
\hline
\end{array}
$$

4.10 Attributes and Attribute Groups

The interpretation of an attribute. `attributeOperator` and `attributeName` determines the set of possible attributes in the substrate relationship table. `reverseFlag` and `expressionConstraintValue` determine the set of candidate targets (if `reverseFlag` is absent) or `subjects` (if `reverseFlag` is present).

`cardinality` determines the minimum and maximum matches. In all cases, only a subset of the subjects (targets if `reverseFlag` is present) in the substrate relationship table will be returned in the interpretation.

```
attribute = [cardinality] [reverseFlag] [attributeOperator] attributeName
       (comparisonOperator concreteValue / "=" expressionConstraintValue )
attributeGroup = "{" attributeSet "}"
attributeSet = attribute [binaryOperator  attributeSet] / "(" attributeSet ")"
cardinality = "[" nonNegativeIntegerValue to (nonNegativeIntegerValue / many) "]"
attributeOperator = descendantOrSelfOf / descendantOf
comparisonOperator = "=" / "!" ws "=" /
    ("n"/"N") ("o"/"O") ("t"/"T") ws "=" / "<=" / "<" / ">=" /  ">"
```

4.11 AttributeName and AttributeSubject

i_attributeName verifies that *conceptReference* is a valid concept in the substrate and interprets it as a set of (exactly one) *ATTRIBUTE*

i_attributeSubject interprets the constraintOperator, if any in the context of the attribute name and interprets it as a set of *ATTRIBUTE*s

Question: Should we add any additional validation checks?

$$attributeName == conceptReference$$

$$attributeOperator == \{descendantOrSelfOf, descendantOf\}$$
$$attributeSubject == attributeOperator[0\,..\,1] \times attributeName$$

$$i\_attributeName : Substrate \pfun attributeName \pfun Result$$
$$i\_attributeSubject : Substrate \pfun attributeSubject \pfun Result$$

$$\forall\, ss : Substrate;\ an : attributeName \bullet i\_attributeName\ ss\ an = i\_conceptReference\ ss\ an$$

$$\forall\, ss : Substrate;\ as : attributeSubject \bullet i\_attributeSubject\ ss\ as = i\_constraintOperator\ ss\ (first\ as)\ (i\_attributeName\ ss\ (second\ as))$$

4.12 Attribute

`attribute` consists of an optional `cardinality` and an `attributeConstraint`.

$$unlimitedNat ::= num\langle\langle \mathbb{N} \rangle\rangle \mid many$$
$$cardinality == \mathbb{N} \times unlimitedNat$$

attribute

$card : cardinality[0 \ldots 1]$

$attw : attributeConstraint$

attributeConstraint is either an attribute expression constraint or a concrete value constraint.

$$attributeConstraint ::= aec\langle\langle attributeExpressionConstraint\rangle\rangle \mid acvc\langle\langle attributeConcreteValueConstraint\rangle\rangle$$

`attributeExpressionEonstraint` is a combination of a subject constraint (target if reverse flag is true) and an expression constraint value whose interpretation yields a set of SCTID's that filter the target (subject if reverse flag).

$[reverseFlag]$

attributeExpressionConstraint

$a : attributeSubject$

$rf : reverseFlag[0 \ldots 1]$

$cs : expressionConstraintValue$

A concrete value constraint is the combination of a subject constraint and a comparison operator/concrete value whose interpretation yields a set of subject ids. The intersection of the subject and concrete value interpretation is the interpretation of `attributeConcreteValueConstraint`

$$comparisonOperator ::= eq \mid neq \mid gt \mid ge \mid lt \mid le$$

attributeConcreteValueConstraint

$a : attributeSubject$

$op : comparisonOperator$

$v : concreteValue$

The interpretation of an attribute is the interpretation of the cardinality applied against the underlying attribute constraint, producing a set of SCTID's or an error condition

The interpretation of an attribute constraint is the interpretation of either the attribute expression constraint or the concrete expression constraint that produces a set of Quads or an error condition.

The actual interpretations if attribute expression and concrete value are in Section 5

$$
\begin{array}{l}
i\_attribute : \\
\qquad Substrate \pfun attribute \pfun Result \\
i\_attributeConstraint : \\
\qquad Substrate \pfun attributeConstraint \pfun Quads \\
i\_attributeExpressionConstraint : \\
\qquad Substrate \pfun attributeExpressionConstraint \pfun Quads \\
i\_attributeConcreteValueConstraint : \\
\qquad Substrate \pfun attributeConcreteValueConstraint \pfun Quads \\
\hline
\forall\, ss : Substrate;\ a : attribute \bullet \\
\quad i\_attribute\ ss\ a = i\_cardinality\ a.card\ (i\_attributeConstraint\ ss\ a.attw) \\
\forall\, ss : Substrate;\ aw : attributeConstraint \bullet i\_attributeConstraint\ ss\ aw = \\
\quad \textbf{if}\ aec^{\sim} aw \in attributeExpressionConstraint \\
\qquad \textbf{then}\ i\_attributeExpressionConstraint\ ss\ (aec^{\sim} aw) \\
\quad \textbf{else if}\ acvc^{\sim} aw \in attributeConcreteValueConstraint \\
\qquad \textbf{then}\ i\_attributeConcreteValueConstraint\ ss\ (acvc^{\sim} aw) \\
\quad \textbf{else}\ qerror\ unknownOperation \\
\forall\, ss : Substrate;\ aec : attributeExpressionConstraint;\ eca : expressionConstraintArgs \mid \\
\quad eca.atts = i\_attributeSubject\ ss\ aec.a \land \\
\quad eca.rf = aec.rf \land \\
\quad eca.subjOrTarg = i\_expressionConstraintValue\ ss\ aec.cs \bullet \\
\quad i\_attributeExpressionConstraint\ ss\ aec = i\_attributeExpression\ ss\ eca \\
\forall\, ss : Substrate;\ awc : attributeConcreteValueConstraint;\ aca : concreteConstraintArgs \mid \\
\quad aca.atts = i\_attributeSubject\ ss\ awc.a \land \\
\quad aca.op = awc.op \land \\
\quad aca.t = awc.v \bullet \\
\quad i\_attributeConcreteValueConstraint\ ss\ awc = i\_concreteAttributeConstraint\ ss\ aca
\end{array}
$$

4.13 AttributeSet and AttributeGroup

An `attributeGroup` is an `attributeSet`. An `attributeSet` consists of a sequence of one or more attribute constraints joined by `binaryOperator`s.

$$
\begin{array}{l}
binaryOperator ::= conjunction \mid disjunction \mid exclusion \\
attributeGroup == attributeSet \\
attributeSet == attribute \times \mathrm{seq}(binaryOperator \times attribute)
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_attributeGroup : Substrate \pfun attributeGroup \pfun Result \\
i\_attributeSet : Substrate \pfun attributeSet \pfun Result \\
\hline
\forall\, ss : Substrate;\ ag : attributeGroup \bullet i\_attributeGroup\ ss\ ag = \\
\quad i\_attributeSet\ ss\ ag \\
\forall\, ss : Substrate;\ a : attributeSet \bullet i\_attributeSet\ ss\ a = \\
\quad result\ (evalCmpndAtt\ ss\ (i\_groupedAttribute\ ss\ (first\ a))\ (second\ a)) \\
\hline
\end{array}
$$

4.14 Compound attribute evaluation

The left-to-right evaluation of `attributeSet`. The interpretation takes a *lhs* as a function from *SCTID* to *GROUP* and a *rhs* which is sequence of operator/attribute tuples and recursively interprets the *lhs* and interpretation of the head of the *rhs*.

$$
\begin{array}{l}
\textit{evalCmpndAtt} : \\
\qquad \textit{Substrate} \to \textit{IDGroups} \to \mathrm{seq}(\textit{binaryOperator} \times \textit{attribute}) \to \textit{IDGroups} \\
\\
\textit{gintersect}, \textit{gunion}, \textit{gminus}, \textit{gfirstError} : \\
\qquad \textit{IDGroups} \to \textit{IDGroups} \to \textit{IDGroups} \\
\hline
\forall\, ss : \textit{Substrate};\ \textit{lhs} : \textit{IDGroups};\ \textit{rhs} : \mathrm{seq}(\textit{binaryOperator} \times \textit{attribute}) \bullet \\
\textit{evalCmpndAtt ss lhs rhs} = \\
\textbf{if } \textit{rhs} = \langle\rangle \\
\quad \textbf{then } \textit{lhs} \\
\textbf{else if } \textit{first}\,(\textit{head rhs}) = \textit{conjunction} \\
\textbf{then } \textit{evalCmpndAtt ss}\,(\textit{gintersect lhs}(\textit{i\_groupedAttribute ss}\,(\textit{second}\,(\textit{head rhs}))))\,(\textit{tail rhs}) \\
\textbf{else if } \textit{first}\,(\textit{head rhs}) = \textit{disjunction} \\
\textbf{then } \textit{evalCmpndAtt ss}\,(\textit{gunion lhs}(\textit{i\_groupedAttribute ss}\,(\textit{second}\,(\textit{head rhs}))))\,(\textit{tail rhs}) \\
\textbf{else if } \textit{first}\,(\textit{head rhs}) = \textit{exclusion} \\
\textbf{then } \textit{evalCmpndAtt ss}\,(\textit{gminus lhs}(\textit{i\_groupedAttribute ss}\,(\textit{second}\,(\textit{head rhs}))))\,(\textit{tail rhs}) \\
\textbf{else } \textit{gerror unknownOperation} \\
\\
\forall\, a, b, r : \textit{IDGroups} \mid \\
\quad r = \textbf{if } \textit{gerror}^{\sim} a \in \textit{ERROR} \lor \textit{gerror}^{\sim} b \in \textit{ERROR} \textbf{ then } \textit{gfirstError } a\ b \\
\quad \textbf{else } \textit{gv}\,(\textit{gv}^{\sim} a \cap \textit{gv}^{\sim} b) \bullet \\
\quad \textit{gintersect } a\ b = r \\
\\
\forall\, a, b, r : \textit{IDGroups} \mid \\
\quad r = \textbf{if } \textit{gerror}^{\sim} a \in \textit{ERROR} \lor \textit{gerror}^{\sim} b \in \textit{ERROR} \textbf{ then } \textit{gfirstError } a\ b \\
\quad \textbf{else } \textit{gv}\,(\textit{gv}^{\sim} a \cup \textit{gv}^{\sim} b) \bullet \\
\quad \textit{gunion } a\ b = r \\
\\
\forall\, a, b, r : \textit{IDGroups} \mid \\
\quad r = \textbf{if } \textit{gerror}^{\sim} a \in \textit{ERROR} \lor \textit{gerror}^{\sim} b \in \textit{ERROR} \textbf{ then } \textit{gfirstError } a\ b \\
\quad \textbf{else } \textit{gv}\,(\textit{gv}^{\sim} a \setminus \textit{gv}^{\sim} b) \bullet \\
\quad \textit{gminus } a\ b = r
\end{array}
$$

$$
\begin{array}{l}
\textit{i\_groupedAttribute} : \\
\qquad \textit{Substrate} \to \textit{attribute} \to \textit{IDGroups} \\
\hline
\forall\, ss : \textit{Substrate};\ a : \textit{attribute} \bullet \\
\quad \textit{i\_groupedAttribute ss } a = \textit{i\_groupCardinality}\,(\textit{i\_attributeConstraint ss } a.\textit{attw})\ a.\textit{card}
\end{array}
$$

4.15 Group Cardinality

The interpretation of cardinality within a group impose additional constraints:

- $[0\,..\,n]$ – the set of all substrate concept codes that have at least one group (entry) in the substrate relationships and, at most n matching entries in the same group

- $[0\,..\,0]$ – the set of all substrate concept codes that have at least one group (entry) in the substrate relationships and *no* matching entries
- $[1..*]$ – (default) at least one matching entry in the substrate relationships
- $[m_1\,..\,n_1]op[m_2\,..\,n_2]...$ – set of substrate concept codes where there exists at least one group where all conditions are simultaneously true

The interpretation of a grouped cardinality is a function from a set of SCTID's to the groups in which they were qualified.

The algorithm below partitions the input set of Quads by group and validates the cardinality on a per-group basis. Groups that pass are returned

TODO: This assumes that q.t is always type object. It doesn't say what to do if it is concrete **TODO:** the *gresult* function seems to express what is described below more simply

$$
\begin{array}{l}
i\_groupCardinality : \\
\qquad Quads \rightarrow cardinality[0\,..\,1] \rightarrow IDGroups \\
\hline
\forall\, quads : Quads;\ oc : cardinality[0\,..\,1];\ uniqueGroups : \mathbb{P}\, GROUP; \\
\qquad quadsByGroup : GROUP \nrightarrow \mathbb{P}\, Quad \mid \\
\quad uniqueGroups = \{q : qids\ quads \bullet q.g\} \land \\
\quad quadsByGroup = \{g : uniqueGroups;\ q : \mathbb{P}\, Quad \mid \\
\qquad q = \{e : qids\ quads \mid e.g = g\} \bullet g \mapsto (evalCardinality\ oc\ q)\} \bullet \\
i\_groupCardinality\ quads\ oc = \\
\quad gv\,\{sctid : SCTID;\ groups : \mathbb{P}\, GROUP \mid sctid \in \{q : \bigcup(\mathrm{ran}\ quadsByGroup) \bullet \\
\qquad \textbf{if}\ qidd\ quads = subjects\ \textbf{then}\ q.s\ \textbf{else}\ object^{\sim} q.t\} \land \\
\quad groups = \{g : \mathrm{dom}\ quadsByGroup \mid (\exists\, q : quadsByGroup\ g \bullet \\
\qquad sctid = \textbf{if}\ qidd\ quads = subjects\ \textbf{then}\ q.s\ \textbf{else}\ object^{\sim} q.t)\} \bullet \\
\qquad sctid \mapsto groups\}
\end{array}
$$

4.16 Cardinality

Interpretation: *cardinality* is tested against a set of quads with the following rules:

1. Errors are propagated
2. No cardinality or passing cardinality returns the subjects / targets of the set of quads
3. Otherwise return an empty set

$$
\begin{array}{l}
i\_cardinality : \\
\qquad cardinality[0\,..\,1] \rightarrow Quads \rightarrow Result \\
\hline
\forall\, oc : cardinality[0\,..\,1];\ qr : Quads \bullet \\
i\_cardinality\ oc\ qr = \\
\quad \textbf{if}\ qerror^{\sim} qr \in ERROR\ \textbf{then}\ result\,(gresult\ qr) \\
\quad \textbf{else}\ result\,(gresult\,(qv\,(evalCardinality\ oc\,(qids\ qr), qidd\ qr)))
\end{array}
$$

evalCardinality Evaluate the cardinality of a set, returning the set if it meets the constraints, otherwise return the empty set.

$$
\begin{array}{|l}
\hline
evalCardinality : cardinality[0 \mathinner{..} 1] \to \mathbb{P}\, Quad \to \mathbb{P}\, Quad \\
\hline
\forall\, oc : cardinality[0 \mathinner{..} 1];\ s : \mathbb{P}\, Quad \bullet evalCardinality\ oc\ s = \\
\quad \textbf{if } \#oc = 0 \textbf{ then } s \\
\quad \textbf{else if } (\#s \geq first\,(head\ oc)) \land \\
\qquad (second\,(head\ oc) = many \lor num^{\sim}(second\,(head\ oc)) \leq \#s) \\
\quad \textbf{then } s \\
\quad \textbf{else } \emptyset \\
\hline
\end{array}
$$

5 Substrate Interpretations

This section defines the interpretations that are realized against the substrate.

5.1 AttributeExpressionConstraint

expressionConstraintArgs carries the arguments necessary to evaluate an attribute expression

- **rf** – If present, return subjects matching attribute/target subjects
- **subjOrTarg** – Set of subject or target SCTIDS (or error)
- **atts** – Set of attributes to evaluate

$$
\begin{array}{|l}
\hline
expressionConstraintArgs \\
\hline
rf : reverseFlag[0 \mathinner{..} 1] \\
subjOrTarg : Result \\
atts : Result \\
\hline
\end{array}
$$

Evaluate an attribute expression, returning the set of quads in the substrate relationship table with the supplied subject / attribute if *rf* is absent and attribute / target if *rf* is present. The interpretation returns an error there is an error in either the subject/targets or attributes. **TODO:** We don't say what to do if the target (t) is concrete

$$
\begin{array}{l}
i\_attributeExpression : \\
\qquad Substrate \to expressionConstraintArgs \to Quads \\
\hline
\forall\, ss : Substrate;\ args : expressionConstraintArgs; \\
\qquad sti : \mathbb{P}\, TARGET;\ atts : \mathbb{P}\, ATTRIBUTE \mid \\
sti = ok^{\sim} args.subjOrTarg \land atts = ok^{\sim} args.atts \bullet \\
i\_attributeExpression\ ss\ args = \\
\qquad \textbf{if } error^{\sim}(bigunion\{args.subjOrTarg, args.atts\}) \in ERROR \\
\qquad \textbf{then} \\
\qquad\qquad qfirstError\{args.subjOrTarg, args.atts\} \\
\qquad \textbf{else if } \neg\ (sti \cup atts) \subseteq ss.c \\
\qquad \textbf{then} \\
\qquad\qquad qerror\ unknownConceptReference \\
\qquad \textbf{else if } \#args.rf = 0 \\
\qquad \textbf{then} \\
\qquad\qquad qv(\{s : SUBJECT;\ t : sti;\ a : atts;\ g : GROUP;\ re : ss.r \mid \\
\qquad\qquad\qquad re.t = t \land re.a = a \bullet re\}, subjects) \\
\qquad \textbf{else} \\
\qquad\qquad qv(\{s : sti;\ a : atts;\ t : TARGET;\ g : GROUP;\ re : ss.r \mid \\
\qquad\qquad\qquad re.s = object^{\sim} t \land re.a = a \bullet re\}, targets)
\end{array}
$$

5.2 ConcreteAttributeConstraint

```
concreteValue =  QM stringValue QM / "#" numericValue
stringValue = 1*(anyNonEscapedChar / escapedChar)
numericValue = decimalValue / integerValue
integerValue = ( ["-"/"+"] digitNonZero *digit ) / zero
```

The mechanism for interpreting *concreteValue* is not fully specified at this point. Much of the rest of the machinery is focused around interpreting constraints in the context of quads - subject, attribute, target and group, so the function is currently defined as returning a *Quads*, but another type or interpretation may be in order.

$$concreteValue ::= stringValue \mid integerValue \mid realValue$$

concreteConstraintArgs The arguments to the concrete value function.

- **atts** – list of attributes to test
- **op** – operator
- **t** – value to test against

Note: *reverseFlag* has no meaning for concrete constraints.

$$
\begin{array}{l}
concreteConstraintArgs \\
\hline
atts : Result \\
op : comparisonOperator \\
t : concreteValue \\
\hline
\end{array}
$$

$$
\begin{array}{|l}
\hline
i\_concreteAttributeConstraint : \\
\quad Substrate \to concreteConstraintArgs \to Quads \\
\hline
\end{array}
$$

5.3 ConstraintOperator

```
constraintOperator =
        descendantOrSelfOf / descendantOf /  ancestorOrSelfOf / ancestorOf
```

Interpretation: Apply the substrate descendants (*descs*) or ancestors (*ancs*) function to a set of SCTID's in the supplied *Result*. Error conditions are propagated.

$$
\begin{array}{|l}
\hline
i\_constraintOperator : \\
\quad Substrate \pfun constraintOperator[0\,..\,1] \pfun Result \pfun Result \\
completeFun : (SCTID \pfun \mathbb{P}\, SCTID) \to SCTID \to \mathbb{P}\, SCTID \\
\hline
\forall\, ss : Substrate;\ oco : constraintOperator[0\,..\,1];\ refset : Result \bullet \\
i\_constraintOperator\ ss\ oco\ refset = \\
\quad \textbf{if}\ error^{\sim} refset \in ERROR \lor \#oco = 0 \\
\quad\quad \textbf{then}\ refset \\
\quad \textbf{else if}\ head\ oco = descendantOrSelfOf \\
\quad\quad \textbf{then}\ ok(\bigcup\{id : ids\ refset \bullet completeFun\ ss.descs\ id \cup \{id\}\}) \\
\quad \textbf{else if}\ head\ oco = descendantOf \\
\quad\quad \textbf{then}\ ok(\bigcup\{id : ids\ refset \bullet completeFun\ ss.descs\ id \setminus \{id\}\}) \\
\quad \textbf{else if}\ head\ oco = ancestorOrSelfOf \\
\quad\quad \textbf{then}\ ok(\bigcup\{id : ids\ refset \bullet completeFun\ ss.ancs\ id \cup \{id\}\}) \\
\quad \textbf{else if}\ head\ oco = ancestorOf \\
\quad\quad \textbf{then}\ ok(\bigcup\{id : ids\ refset \bullet completeFun\ ss.ancs\ id \setminus \{id\}\}) \\
\quad \textbf{else}\ error\ unknownOperation \\
\forall\, f : (SCTID \pfun \mathbb{P}\, SCTID);\ id : SCTID \bullet completeFun\ f\ id = \\
\quad \textbf{if}\ id \in \mathrm{dom}\, f\ \textbf{then}\ f\ id\ \textbf{else}\ \emptyset \\
\hline
\end{array}
$$

5.4 FocusConcept

```
focusConcept = [memberOf] conceptReference
```

Interpretation: focusConcept is interpreted as the interpretation of conceptReference itself or, if memberOf is specified, the interpretation of the memberOf function applied to the conceptReference

$$focusConcept ::= mo\langle\langle conceptReference\rangle\rangle \mid cr\langle\langle conceptReference\rangle\rangle$$

$$
\begin{array}{l}
i\_focusConcept : Substrate \rightarrow focusConcept \rightarrow Result \\
\hline
\forall\, ss : Substrate;\ fc : focusConcept \bullet \\
i\_focusConcept\ ss\ fc = \\
\quad \textbf{if}\ cr^{\sim} fc \in conceptReference \\
\quad\quad \textbf{then}\ i\_conceptReference\ ss\ (cr^{\sim} fc) \\
\quad \textbf{else if}\ mo^{\sim} fc \in conceptReference \\
\quad\quad \textbf{then}\ i\_memberOf\ ss\ (mo^{\sim} fc) \\
\quad \textbf{else}\ error\ unknownOperation
\end{array}
$$

5.4.1 memberOf

Interpretation: memberOf is interpreted as application *refset* function to the conceptReference SCTID or an error if the SCTID isn't in the domain of the *refset* function.

$$
\begin{array}{l}
i\_memberOf : Substrate \rightarrow conceptReference \rightarrow Result \\
\hline
\forall\, ss : Substrate;\ cr : conceptReference \bullet \\
i\_memberOf\ ss\ cr = \\
\quad (\textbf{let}\ sctid == toSCTID\ cr \bullet \\
\quad\quad \textbf{if}\ sctid \notin \mathrm{dom}\, ss.refset\ \textbf{then}\ error\ unknownRefsetId \\
\quad \textbf{else}\ ok\ (ss.refset\ sctid))
\end{array}
$$

5.5 ConceptReference

```
conceptReference = conceptId [ "|" Term "|"]
conceptId = sctId
```

Interpretation: conceptReference is interpreted as *SCTID* if it is a member the list of concepts, c in the substrate, otherwise as an *error*.

[*conceptReference*]

$$
\begin{array}{l}
toSCTID : conceptReference \rightarrow SCTID \\
i\_conceptReference : Substrate \rightarrow conceptReference \rightarrow Result \\
\hline
\forall\, ss : Substrate;\ c : conceptReference \bullet i\_conceptReference\ ss\ c = \\
\quad \textbf{if}\ (toSCTID\ c) \in ss.c\ \textbf{then}\ ok\{(toSCTID\ c)\} \\
\quad \textbf{else}\ error\ unknownConceptReference
\end{array}
$$

6 Glue and Helper Functions

This section carries various type transformations and error checking functions

6.1 Types

- **direction** – an indicator whether a collection of quads was determined as subject to target (*subjects*) or target to subject (*targets*)
- **Quads** – a collection of *Quad*s or an error condition. If it is a collection *Quad*s, it also carries a direction indicator that determines whether it represents a set of subjects or targets.
- **IDGroups** – a map from *SCTID*s to the *GROUP* they were in when they passed if successful, otherwise an error indication.

$direction ::= subjects \mid targets$
$Quads ::= qv\langle\langle \mathbb{P}\, Quad \times direction \rangle\rangle \mid qerror\langle\langle ERROR \rangle\rangle$

$IDGroups ::= gv\langle\langle SCTID \pfun \mathbb{P}\, GROUP \rangle\rangle \mid gerror\langle\langle ERROR \rangle\rangle$

6.2 Result transformations

- **ids** – return the set of *SCTID*s in a result or the empty set if there is an error
- **qids** – return the set of *Quads* in a quads result or an empty set if there is an error
- **qidd** – return the direction of a *Quads* result. Undefined if error
- **gids** – return the *SCTID* to group map in an id group or an empty map if there is error
- **gresult** – convert a set of quads int a set of id groups
- **result** – convert a set of id groups into a simple result.

$$
\begin{array}{l}
ids : Result \to \mathbb{P}\, SCTID \\
qids : Quads \to \mathbb{P}\, Quad \\
qidd : Quads \nrightarrow direction \\
gids : IDGroups \to SCTID \to \mathbb{P}\, GROUP \\
gresult : Quads \to IDGroups \\
result : IDGroups \to Result \\
\hline
\forall\, r : Result \bullet ids\, r = \\
\quad \textbf{if } error^{\sim} r \in ERROR \textbf{ then } \emptyset \\
\quad \textbf{else } ok^{\sim} r \\
\\
\forall\, q : Quads \bullet qids\, q = \\
\quad \textbf{if } qerror^{\sim} q \in ERROR \textbf{ then } \emptyset \\
\quad \textbf{else } first\, (qv^{\sim} q) \\
\\
\forall\, q : Quads \bullet qidd\, q = \\
\quad second\, (qv^{\sim} q) \\
\\
\forall\, g : IDGroups \bullet gids\, g = \\
\quad \textbf{if } gerror^{\sim} g \in ERROR \textbf{ then } \emptyset \\
\quad \textbf{else } gv^{\sim} g \\
\\
\forall\, q : Quads \bullet gresult\, q = \\
\quad \textbf{if } qerror^{\sim} q \in ERROR \textbf{ then } gerror(qerror^{\sim} q) \\
\quad \textbf{else if } qidd\, q = subjects \\
\qquad \textbf{then } gv\, \{s : SCTID \mid (\exists\, qr : qids\, q \bullet s = qr.s) \bullet \\
\qquad s \mapsto \{qr : qids\, q \bullet qr.g\}\} \\
\quad \textbf{else} \\
\qquad gv\, \{t : SCTID \mid (\exists\, qr : qids\, q \bullet t = object^{\sim} qr.t) \bullet \\
\qquad t \mapsto \{qr : qids\, q \bullet qr.g\}\} \\
\\
\forall\, g : IDGroups \bullet result\, g = \\
\quad \textbf{if } gerror^{\sim} g \in ERROR \textbf{ then } error(gerror^{\sim} g) \\
\quad \textbf{else } ok(\mathrm{dom}(gv^{\sim} g))
\end{array}
$$

Definition of the various functions that are performed on the result type.

- **firstError** – take a set of *Result*s with one or more errors and return an aggregation / summary as a *Result*. (Not fully defined)
- **qfirstError** – take a set of *Result*s with one or more errors and return an aggregation / summary as a *Quads* instance. (not fully defined)

- **union** – return the union of two *Result SCTID*s or *ERROR* if either of them have an error.
- **intersect** – return the intersection of two *Result SCTID*s or *ERROR* if either of them have an error.

- **minus** – return the set of *SCTID*s in the first set of results that are not in the second or *ERROR* if either of them have an error.
- **bigunion** – return the union of a set *Result SCTID*s or *ERROR* if any of the results in the results in the set have an error.
- **bigintersect** – return the intersection of a set *Result SCTID*s or *ERROR* if any of the results in the set have an error.

firstError is not fully defined – it takes a set of results with one or more errors and returns an aggregation of all of the errors.

$$
\begin{array}{l}
union, intersect, minus : Result \to Result \to Result \\
bigunion, bigintersect : \mathbb{P}\, Result \to Result \\
firstError : \mathbb{P}\, Result \pfun Result \\
qfirstError : \mathbb{P}\, Result \pfun Quads \\
\hline
\forall\, x, y : Result \bullet union\, x\, y = \\
\quad \textbf{if } error^{\sim} x \in ERROR \textbf{ then } x \\
\quad \textbf{else if } error^{\sim} y \in ERROR \textbf{ then } y \\
\quad \textbf{else } ok\, ((ok^{\sim} x) \cup (ok^{\sim} y)) \\
\\
\forall\, x, y : Result \bullet intersect\, x\, y = \\
\quad \textbf{if } error^{\sim} x \in ERROR \textbf{ then } x \\
\quad \textbf{else if } error^{\sim} y \in ERROR \textbf{ then } y \\
\quad \textbf{else } ok\, ((ok^{\sim} x) \cap (ok^{\sim} y)) \\
\\
\forall\, x, y : Result \bullet minus\, x\, y = \\
\quad \textbf{if } error^{\sim} x \in ERROR \textbf{ then } x \\
\quad \textbf{else if } error^{\sim} y \in ERROR \textbf{ then } y \\
\quad \textbf{else } ok\, ((ok^{\sim} x) \setminus (ok^{\sim} y)) \\
\\
\forall\, rs : \mathbb{P}\, Result \bullet bigunion\, rs = \\
\quad \textbf{if } \exists\, r : rs \bullet error^{\sim} r \in ERROR \textbf{ then } firstError\, rs \\
\quad \textbf{else } ok\, (\bigcup \{ r : rs \bullet ids\, r \}) \\
\\
\forall\, rs : \mathbb{P}\, Result \bullet bigintersect\, rs = \\
\quad \textbf{if } \exists\, r : rs \bullet error^{\sim} r \in ERROR \textbf{ then } firstError\, rs \\
\quad \textbf{else } ok\, (\bigcap \{ r : rs \bullet ids\, r \})
\end{array}
$$

7 Appendix 1

Representing optional elements of type T. Representing it as a sequence allows us to determine absence by $\# T = 0$ and the value by $head\ T$.

$$T[0\,..\,1] == \{ s : \mathrm{seq}\ T \mid \# s \leq 1 \}$$